
Captions: 

ESM 1. Scatterplots of the AUC24 prediction versus the prediction on the full pharmacokinetic curve 

for the limited sampling strategies using sampling (a) at 2, 4 and 6 hours after dosing, (b) at 2 and 4 

hours after dosing, and (c) at 2 hours after dosing. 

AUC24 area under the concentration time curve from 0 to 24 hours after dosing, PK pharmacokinetic 

ESM 2. Prediction of the AUC24 using the 2 and 4 hour sampling strategy versus AUC24 prediction on 

the full pharmacokinetic curve for (a) the final model without acetyl-isoniazid data, and (b) the re-

estimated model without mixture component. 

AUC24 area under the concentration time curve from 0 to 24 hours after dosing, PK pharmacokinetic 

ESM 3. NONMEM control stream of the population pharmacokinetic model of isoniazid and acetyl-
isoniazid 

ESM 3 
 
$SUBROUTINE ADVAN5 
 
$MODEL 
COMP=(DOSE,DEFDOSE) 
COMP=(CENTRAL,DEFOBS) 
COMP=(TRANSIT1) 
COMP=(TRANSIT2) 
COMP=(TRANSIT3) 
COMP=(TRANSIT4) 
COMP=(PERIPH) 
COMP=(LIVER) 
COMP=(ACINH) 
 
$MIX 
NSPOP=2 
P(1) = THETA(8) 
P(2) = 1-THETA(8) 
 
$PK 
FFM2= FFM 
IF(FFM.EQ.-99) FFM2 = 45 
 
FCL = (FFM2/45)**0.75 
FV  = (FFM2/45)**1 
 
TVV = THETA(1) * FV 
TVCLF  = THETA(2) * FCL 
TVCLS = THETA(9) * FCL 
TVKA  = THETA(3) 
TVV2 = THETA(4) * FV 
TVQ = THETA(5) * FCL 
TVVM = THETA(6) * FV 
TVCLM = THETA(7) * FCL 
TVCLO = THETA(10)* FCL 
 
V    = TVV*EXP(ETA(1)) 
IF(MIXNUM.EQ.1) CL  = TVCLF*EXP(ETA(5)) 
IF(MIXNUM.EQ.2) CL  = TVCLS*EXP(ETA(5)) 
MIX    = MIXNUM 
KA    = TVKA*EXP(ETA(2)) 
V2    = TVV2 
Q    = TVQ 
VM    = TVVM*EXP(ETA(4)) 
CLM    = TVCLM*EXP(ETA(3)) 
CLO    = TVCLO*EXP(ETA(6) 
 
VH  = 1 * FV 
QH  = 0.55 * 90 * FCL 
FU  = 0.9 
 



CLINT  = CL+CLO 
EH  = ( CLINT * FU ) / ( ( CLINT * FU ) + QH ) 
CLH  = EH * QH 
 
EH_ACINH = ( CL * FU ) / ( ( CL * FU ) + QH ) 
CL_ACINH = EH_ACINH * QH 
 
K13 = KA 
K34 = KA 
K45 = KA 
K56 = KA 
K68 = KA 
K27 = Q/V 
K72 = Q/V2 
K82  = (QH*(1-EH))/VH 
K28  = QH/V 
K89 = CL_ACINH/VH 
K90 = CLM/VM 
K80 = CLO/VH 
 
$ERROR 
 
IF(STDY.EQ.1) THEN 
 ALT_BLQ = 1 
ELSE 
 ALT_BLQ = 0 
ENDIF 
 
IF(CMT.EQ.2) THEN 
 ERR_BLQ = ( ALT_BLQ * ERR(3) ) + ( ( 1-ALT_BLQ ) * ERR(4) ) 
 ERRORS= ERR(1) 
 IPRED=A(2)/V 
ELSE 
 ERR_BLQ = ERR(5) 
 ERRORS= ERR(2) 
 IPRED=A(9)/VM 
ENDIF 
 
IF(BLQ.EQ.0) THEN 
 Y = IPRED + IPRED*(ERRORS) 
ELSE  
 Y = IPRED + IPRED*(ERRORS) + ERR_BLQ 
ENDIF 
 
IRES = DV - IPRED 
IWRES = IRES/IPRED 
 
$THETA 
(0, 57.5) ;1 V  
(0, 32.7) ;2 CL FAST 
(0, 5.42) ;3 KA 
(0, 18.7) ;4 V2 



(0, 2.48) ;5 Q 
(0, 39.2) ;6 VM 
(0, 6.65) ;7 CLM 
(0, 0.434,1) ;8 PROP FAST 
(0, 4.31) ;9 CL SLOW 
(0, 12.1) ;10 CL OTHER 
 
$OMEGA  
 0.0697 ;1 IIV V 
 0.692 ;2 IIV KA 
 0 FIX  ;3 IIV F 
 0.135 ;4 IIV CLM 
 0.0107 ;5 IIV VM 
$OMEGA BLOCK(2) 
 0.331  ;6 IIV CL 
 0.0312 0.144 ;7 IIV CLO  
 
$SIGMA BLOCK(2) 
 0.141 ;1 Proportional error parent 
 0.0525 0.0544 ;2 Proportional error metabolite 
$SIGMA 
 0.000002 FIX  ;3 additive BLQ error parent analytical method 1 
 0.00000008 FIX  ;4 additive BLQ error parent analytical method 2 
 0.0000002 FIX  ;5 additive BLQ error metabolite 
 
$ESTIMATION METH=1 INTER PRINT=1 


